RCSL : 44-RT-1050
Author : Børge Thøgersen
Edited : June 1975

DISK PROGRAM LOAD I

Keywords : Program Load, High Speed Devices ROM 026+ROM 027, ROM 028+ROM 029.

Abstract : The Disk Program Load is capable of making an autoload from Disk cartridge drives connected to the DCC 701 Controller and the DCA 701 Adaptor. The Disk Program Load is also able to autoload from block-oriented devices of e.g. magnetic tape type.

CONTENTS PAGE

1. SCOPE

The Disk Program Load program is a 32-word program which is stored in two integrated Read Only Memories. The two ROM's are physically placed on the CPU board.

3600 C Models: ROM 026 contains the 32 left bytes and is placed in position 30 on the CPU board.

ROM 027 contains the 32 right bytes and is placed in position 31.

3600 D Models: ROM 028 contains the 32 left bytes and is placed in position 135 on the CPU board.

ROM 029 contains the 32 right bytes and is placed in position 134.

The program is loaded into core by the procedure described below. When the loading is finished, the program is started at location 0.

The Disk Program Load is capable of loading from various high-speed devices.

2. OPERATING PROCEDURE

1. Set device code in dataswitches 10-15.
2. Set dataswitch 0 if load medium is another high-speed device than Disk.
3. Dataswitches 1-9 are Don't-Cares.
4. Make the load medium ready, i.e. the tape on a magnetic tape station must be on line.
5. Start the Program Load : Press Autoload or Autoload-Load on the operator panel, depending on which one is used.

3. PROGRAM DESCRIPTION

If dataswitch 0 is reset, a Disk recalibration is made. The program starts the load medium and makes a jump to location octal 377, which contains a jump to location 377, i.e. when the load has overwritten location 377, the new instruction is executed.

```
                ;    PROGRAM LOAD FROM DISK
                ;    AND OTHER HIGH SPEED DEVICES


                000000  .LOC    0

                000000  DEV=     0

00000 006477    BEG:    READS   1               ; READ SWITCHES
00001 020037            LDA     0    C77        ; ISOLATE DEVICE CODE
00002 123400            AND     1,0
00003 100404            NEG     0,0  SZR
00004 010031    LOOP:   ISZ             OP1     ; COUNT DEVICE CODE
00005 010032            ISZ             OP2     ; INTO ALL IN/OUT
00006 010022            ISZ             OP3     ; INSTRUCTIONS
00007 010025            ISZ             OP4
00010 101404            INC     0,0  SZR        ; DONE
00011 000004            JMP             LOOP    ; NO, INCREMENT AGAIN
00012 125102            MOVL    1,1  SZC        ; DISK
00013 000022            JMP             OP3     ; NO
00014 004030            JSR             SPEC    ; SEEK WITH CLEAR
00015 175000                            175000  ; SEEK INSTRUCTION
00016 004030            JSR             SPEC    ; RECALIBRATE WITH CLEAR
00017 175400                            175400  ; RECALIBRATE INSTRUCTION
00020 004030            JSR             SPEC    ; DISK READY
00021 175000                            175000
00022 006100    OP3:    DOAS    0       DEV     ; START DEVICE WITH
                                                ; RESET INSTRUCTION REGISTER
00023 030027            LDA     2       C377    ; SETUP JMP 377 IN
00024 050377            STA     2       377     ; LOCATION 377
00025 063400    OP4:    SKPBN           DEV     ; BUSY
00026 000022            JMP             OP3     ; NO, START AGAIN
00027 000377    C377:   JMP             377     ; YES, WAIT PAGE ZERO
                                                ; OVERWRITTEN


00030 025400    SPEC:   LDA     1       0,3     ; DISK ROUTINE
00031 065500    OP1:    DOAP    1       DEV     ; RECALIBRATE DISK
00032 064400    OP2:    DIA     1       DEV     ; READ STATUS
00033 131300            MOVS    1,2
00034 133405            AND     1,2  SNR        ; DONE
00035 000032            JMP             OP2     ; NO, WAIT
00036 001401            JMP             1,3     ; RETURN


00037 000077    C77:    77


                .END
```

BEG	000000
C377	000027
C77	000037
DEV	000000
LOOP	000004
OP1	000031
OP2	000032
OP3	000022
OP4	000025
SPEC	000030